

Софийски университет „Св. Климент Охридски”

Факултет по математика и информатика

Курсов проект по

Бази от знания, летен семестър 2024/2025

на тема

**Заявки върху онтология за човешки фенотип –  
Human Phenotype Ontology**

Изготвено от: Цветелина Ненкова Чакърва, факултетен номер: 8MI3400591,  
магистърска програма: Извличане на информация и откриване на знания

Юни, 2025 г.

Гр. София

## Съдържание

|       |                                          |    |
|-------|------------------------------------------|----|
| 1.    | Въведение и основни цели .....           | 3  |
| 2.    | Описание на използваната онтология ..... | 3  |
| 2.1.  | Класове .....                            | 3  |
| 2.2.  | Свойства .....                           | 6  |
| 2.3.  | Допълнителни данни .....                 | 8  |
| 3.    | SPARQL заявки .....                      | 8  |
| 3.1.  | Заявка 1 .....                           | 8  |
| 3.2.  | Заявка 2 .....                           | 10 |
| 3.3.  | Заявка 3 .....                           | 10 |
| 3.4.  | Заявка 4 .....                           | 11 |
| 3.5.  | Заявка 5 .....                           | 12 |
| 3.6.  | Заявка 6 .....                           | 13 |
| 3.7.  | Заявка 7 .....                           | 13 |
| 3.8.  | Заявка 8 .....                           | 14 |
| 3.9.  | Заявка 9 .....                           | 16 |
| 3.10. | Заявка 10 .....                          | 17 |
| 3.11. | Заявка 11 .....                          | 18 |
| 3.12. | Заявка 12 .....                          | 19 |
| 3.13. | Заявка 13 .....                          | 20 |
| 3.14. | Заявка 14 .....                          | 20 |
| 4.    | Заклучение и бъдещо развитие .....       | 22 |
| 5.    | Използвани технологии .....              | 22 |
| 6.    | Файлова структура на проекта .....       | 22 |
| 7.    | Източници .....                          | 22 |


## 1. Въведение и основни цели

Human Phenotype Ontology (HPO) е стандартизирана онтология, която описва фенотипове при хората (съвкупност от външно видимите белези) в контекста на заболявания. Тя предоставя структуриран начин за описание на клинични аномалии. HPO се използва в медицински изследвания и проучвания, както и в биоинформатиката за подпомагане на диагностиката и анализа на заболявания.

Основни цели на проекта са да изследва структурата на онтологията, да представи основната информация, която се съдържа в нея, и да демонстрира приложението ѝ в различни контексти с помощта на SPARQL заявки. Разглеждат се анализиране на различните видове фенотипове и откриване на фенотипове чрез ключови думи.

## 2. Описание на използваната онтология


Проектът е изграден над Human Phenotype Ontology, v2025-05-06. Тя съдържа 903004 твърдения (RDF тройки), поради което са описани част от основните ѝ елементи и са показани частични визуализации.




### 2.1. Класове

Класовете в онтологията могат да се разглеждат като няколко отделни йерархии. Коренът на всяка йерархия е свързан с клас Class чрез връзка от тип type и е свързан с клас All чрез връзка от тип rdfs:subClassOf. Всеки такъв клас има множество подкласове, с които е свързан чрез rdfs:subClassOf връзки. Всички техни подкласове имат свои подкласове, които от своя страна имат подкласове и т.н. – изгражда се дървовидна структура, свързана чрез rdfs:subClassOf връзки. Основните йерархии, представени чрез техните корен, кратко описание и частична визуализация, са:


- **Phenotypic abnormality** (HP:0000118) - Обхваща всички фенотипни аномалии. Директните му подкласове включват: Abnormality of head or neck (HP:0000152), Abnormality of the nervous system (HP:0001333), Abnormality of the skeletal system (HP:0000924) и др.




- **Mode of inheritance** (HP:0000005) – Описва начините на унаследяване на заболявания. Директните му подкласове включват: X-linked dominant inheritance (HP:0001423), Y-linked inheritance (HP:0001450), Sporadic (HP:0003745) и др.




- **Clinical course** (HP:0031797) – Описва развитието на заболяванията във времето. Директните му подкласове включват: Progressive (HP:0003676), Nonprogressive (HP:0003680), Chronic (HP:0011010) и др.



- **Frequency** (HP:0040279) – Описва честотата на проявление на заболяванията. Директните му подкласове включват: Very frequent (HP:0040281), Frequent (HP:0040282), Occasional (HP:0040283) и др.



- **Clinical modifier** (HP:0012823) – Описва модификатори на клиничното представяне. Директните му подкласове включват: Adult onset (HP:0003581), Juvenile onset (HP:0003621), Position (HP:0012830) и др.



## 2.2. Свойства

Свойствата в онтологията образуват йерархии, като някои по-общи свойства са: `rdf:Property`, `owl:AnnotationProperty`, `owl:SymmetricProperty`, `owl:TransitiveProperty`, `owl:ObjectProperty`. Останалите свойства са свързани с тях чрез връзки `rdf:type`, като някои свойства са едновременно от повече от един тип. Например `inverseOf` е едновременно свързано с `rdf:Property` и с `owl:SymmetricProperty` с връзки `rdf:type`:

[inverseOf](#) 

 inverseOf


Types:

`rdf:Property`


`owl:SymmetricProperty`

Свойствата могат да бъдат категоризирани според тяхната функционалност и роля в онтологията. Основните видове свойства са:

- **Структурни свойства** – Определят йерархичната организация на класовете. Например `rdfs:subClassOf`.
- **Релационни свойства** – Дефинират връзки между класове или индивиди.
  - Примери за свойства от тип `owl:ObjectProperty` са: `part_of` (BFO:0000050) - указва, че даден фенотип е част от по-общ фенотип; `occurs_in` (BFO:0000066) – указва контекста, в който се проявява даден фенотип; `has_modifier` (RO:0002180) – използва се за добавяне на допълнителни характеристики към фенотипа като например степен на тежест, честота на проявление и др.



- **Анотационни свойства** – Служат за добавяне на метаданни като описания или синоними. Пример за свойства от тип `owl:AnnotationProperty` са: `has_exact_synonym`, `has_broad_synonym`, `abbreviation` и др.



## 2.3. Допълнителни данни

Заболяванията не са директно представени като класове в самата онтология, а са свързани чрез анотации към външни онтологии и бази данни като OMIM (Online Mendelian Inheritance in Man), Orphanet и MONDO (Monarch Disease Ontology). Това показва, че HPO се фокусира върху фенотиповете, а заболяванията се асоциират с тях чрез външни препратки.

## 3. SPARQL заявки

Предварителни сведения:

- Визуализацията на онтологията в GraphDB се извършва със зададено ограничение за максимален брой от 20 визуализирани връзки, излизащи от един възел, поради което броят на показани възли и връзки в изображенията не винаги съвпада с общия брой налични връзки, които се връщат като резултат от заявките.
- Някои заявки използват префикс PREFIX obo: <http://purl.obolibrary.org/obo/>, който е за означаване на идентификатори (IRI) от хранилището ОВО (Open Biomedical Ontologies), което се използва за биомедицински онтологии.

### 3.1. Заявка, показваща фенотипните аномалии (фенотиповете)

Заявката показва йерархията на класовете за фенотипни аномалии, като извежда подкласовете на класа Phenotypic abnormality (HP:0000118).

Резултат без включен логически извод (Include inferred data in results: OFF):

The screenshot shows a SPARQL query interface. The query is as follows:

```
1 PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
2 PREFIX obo: <http://purl.obolibrary.org/obo/>
3
4 SELECT ?phenotype ?phenotypeLabel
5 WHERE {
6   ?phenotype rdfs:subClassOf obo:HP_0000118 ;
7             rdfs:label ?phenotypeLabel .
8 }
```

On the right side of the query editor, there are icons for saving, opening, linking, and running the query, along with a 'Run' button and a 'keyboard shortcuts' link.

Below the query editor, there are tabs for 'Table', 'Raw response', 'Pivot Table', and 'Google Chart'. The 'Table' tab is selected. To the right of the tabs is a 'Download as' button.


Below the tabs, there is a filter bar with 'Filter query results' and options for 'Compact view' and 'Hide row numbers'. It also shows 'Showing results from 0 to 23 of 23. Query took 0.1s, moments ago.'

The results are displayed in a table with two columns: 'phenotype' and 'phenotypeLabel'.

|   | phenotype      | phenotypeLabel                            |
|---|----------------|-------------------------------------------|
| 1 | obo:HP_0001507 | "Growth abnormality"                      |
| 2 | obo:HP_0000119 | "Abnormality of the genitourinary system" |
| 3 | obo:HP_0000152 | "Abnormality of head or neck"             |
| 4 | obo:HP_0000478 | "Abnormality of the eye"                  |



Резултат с включен логически извод (Include inferred data in results: ON):



The screenshot shows a SPARQL query interface. The query is as follows:

```

1 PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
2 PREFIX obo: <http://purl.obolibrary.org/obo/>
3
4 SELECT ?phenotype ?phenotypeLabel
5 WHERE {
6   ?phenotype rdfs:subClassOf obo:HP_0000118 ;
7             rdfs:label ?phenotypeLabel .
8 }

```

On the right side of the query editor, there are icons for saving, opening, linking, navigating, and a 'Run' button. Below the query editor, there are tabs for 'Table', 'Raw response', 'Pivot Table', and 'Google Chart'. The 'Table' tab is selected. Below the tabs, there is a 'Filter query results' input field and a 'Compact view' checkbox. The results section shows a table with two columns: 'phenotype' and 'phenotypeLabel'. The table contains four rows of results.

|   | phenotype      | phenotypeLabel                 |
|---|----------------|--------------------------------|
| 1 | obo:HP_0000002 | "Abnormality of body height"   |
| 2 | obo:HP_0001507 | "Growth abnormality"           |
| 3 | obo:HP_0000003 | "Multicystic kidney dysplasia" |
| 4 | obo:HP_0000107 | "Renal cyst"                   |

Обяснения на причините за установените различия: Когато логическият извод е изключен, заявката връща само класовете, които са изрично свързани в онтологията със свойството `rdfs:subClassOf` към класа Phenotypic abnormality (HP:0000118). Това води до 23 резултата. Когато логическият извод е включен, свойството `rdfs:subClassOf` е транзитивно, следователно заявката връща както директно свързаните с Phenotypic abnormality (HP:0000118) класове, така и тези класове, които са подкласове на някой подклас на Phenotypic abnormality (HP:0000118). Това води до 18506 резултата.

### 3.2. Заявка, показваща всички свойства от тип owl:ObjectProperty

Заявката показва всички свойства, свързани с owl:ObjectProperty чрез връзка rdf:type.

```
1 PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
2 PREFIX owl: <http://www.w3.org/2002/07/owl#>
3 PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
4
5 SELECT ?property ?propertyLabel
6 WHERE {
7   ?property rdf:type owl:ObjectProperty .
8   OPTIONAL { ?property rdfs:label ?propertyLabel }
9 }
```



keyboard shortcuts

 Table  Raw response  Pivot Table  Google Chart 

Filter query results Compact view ☐ Hide row numbers ☐ Showing results from 0 to 233 of 233. Query took 0.1s, moments ago.

|   | property        | propertyLabel |
|---|-----------------|---------------|
| 1 | obo:BF0_0000050 | "part of"     |
| 2 | obo:BF0_0000050 | "part of"@en  |
| 3 | obo:RO_0002131  | "overlaps"@en |
| 4 | obo:BF0_0000051 | "has part"    |


### 3.3. Заявка, показваща всички свойства, които са едновременно от типовете owl:ObjectProperty, owl:SymmetricProperty и owl:TransitiveProperty

Заявката показва всички свойства, свързани едновременно с owl:ObjectProperty, с owl:SymmetricProperty и с owl:TransitiveProperty чрез връзки rdf:type.

```
1 PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
2 PREFIX owl: <http://www.w3.org/2002/07/owl#>
3 PREFIX rdf: <http://www.w3.org/1999/02/22-rdf-syntax-ns#>
4
5 SELECT ?property ?propertyLabel
6 WHERE {
7   ?property rdf:type owl:ObjectProperty ;
8             rdf:type owl:SymmetricProperty ;
9             rdf:type owl:TransitiveProperty .
10  OPTIONAL { ?property rdfs:label ?propertyLabel }
11 }
```



keyboard shortcuts

 Table  Raw response  Pivot Table  Google Chart 

Filter query results Compact view ☐ Hide row numbers ☐ Showing results from 0 to 3 of 3. Query took 0.1s, moments ago.

|   | property       | propertyLabel            |
|---|----------------|--------------------------|
| 1 | obo:RO_0002082 | "simultaneous with"@en   |
| 2 | obo:RO_0002158 | "shares ancestor with"   |
| 3 | obo:RO_0002159 | "serially homologous to" |

### 3.4. Заявка за търсене по една ключова дума, показваща всички фенотипове, съдържащи “heart”

Етикетите на фенотиповете се филтрират така, че да останат само тези, съдържащи “heart”. Резултатите са сортирани по азбучен ред на етикетите на фенотиповете.

```
1 PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
2
3 SELECT ?phenotype ?phenotypeLabel
4 WHERE {
5   ?phenotype rdfs:label ?phenotypeLabel .
6   FILTER(CONTAINS(LCASE(?phenotypeLabel), "heart"))
7 }
8 ORDER BY ?phenotypeLabel
```

Run

keyboard shortcuts

Table Raw response Pivot Table Google Chart

Download as

Filter query results Compact view ☐ Hide row numbers ☐ Showing results from 0 to 71 of 71. Query took 0.2s, moments ago.

|   | phenotype      | phenotypeLabel                                  |
|---|----------------|-------------------------------------------------|
| 1 | obo:HP_0004307 | "Abnormal anatomic location of the heart"       |
| 2 | obo:HP_4000143 | "Abnormal fetal heart outflow tract"            |
| 3 | obo:HP_4000105 | "Abnormal four chamber view of the fetal heart" |
| 4 | obo:HP_0001627 | "Abnormal heart morphology"                     |

### 3.5. Заявка за търсене по две ключови думи, свързани с логическо “И”, показваща всички фенотипове, съдържащи едновременно “abnormal” и “heart”

Етикетите на фенотиповете се филтрират така, че да останат само тези, съдържащи “abnormal” и “heart”. Резултатите са сортирани по азбучен ред на етикетите на фенотиповете.

```
1 PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
2
3 SELECT ?phenotype ?phenotypeLabel
4 WHERE {
5   ?phenotype rdfs:label ?phenotypeLabel .
6   FILTER(
7     CONTAINS(LCASE(?phenotypeLabel), "abnormal") &&
8     CONTAINS(LCASE(?phenotypeLabel), "heart")
9   )
10 }
11 ORDER BY ?phenotypeLabel
```

Run

keyboard shortcuts

Table Raw response Pivot Table Google Chart Download as

Filter query results Compact view ☐ Hide row numbers ☐ Showing results from 0 to 11 of 11. Query took 0.2s, moments ago.

|   | phenotype      | phenotypeLabel                                  |
|---|----------------|-------------------------------------------------|
| 1 | obo:HP_0004307 | "Abnormal anatomic location of the heart"       |
| 2 | obo:HP_4000143 | "Abnormal fetal heart outflow tract"            |
| 3 | obo:HP_4000105 | "Abnormal four chamber view of the fetal heart" |
| 4 | obo:HP_0001627 | "Abnormal heart morphology"                     |

### 3.6. Заявка за търсене по ключови думи с комбинирана логика - логическо “И” и логическо “ИЛИ”, показваща всички фенотипове, съдържащи едновременно (“abnormal” и “heart”) или (“abnormal” и “cardiac”)

Етикетите на фенотиповете се филтрират така, че да останат само тези, съдържащи (“abnormal” и “heart”) или (“abnormal” и “cardiac”). Резултатите са сортирани по азбучен ред.

```
1 PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
2
3 SELECT ?phenotype ?phenotypeLabel
4 WHERE {
5   ?phenotype rdfs:label ?phenotypeLabel .
6   FILTER(
7     CONTAINS(LCASE(?phenotypeLabel), "abnormal") &&
8     (CONTAINS(LCASE(?phenotypeLabel), "heart") || CONTAINS(LCASE(?phenotypeLabel), "cardiac"))
9   )
10 }
```

Run

keyboard shortcuts

Table Raw response Pivot Table Google Chart Download as

Filter query results Compact view ☐ Hide row numbers ☐ Showing results from 0 to 27 of 27. Query took 0.2s, moments ago.

|   | phenotype      | phenotypeLabel                            |
|---|----------------|-------------------------------------------|
| 1 | obo:HP_0001627 | "Abnormal heart morphology"               |
| 2 | obo:HP_0005120 | "Abnormal cardiac atrium morphology"      |
| 3 | obo:HP_0001654 | "Abnormal heart valve morphology"         |
| 4 | obo:HP_0004307 | "Abnormal anatomic location of the heart" |

### 3.7. Заявка за търсене по ключови думи, която категоризира и сортира намерените фенотипове по броя на съдържащите се в тях ключови думи

Заявката намира всички фенотипове, чиито етикети съдържат поне една от ключовите думи “cardiac”, “muscle”, “contraction”. Чрез BIND се създават две нови променливи - ?matchCount и ?matchType. ?matchCount брои колко от ключовите думи се срещат в етикета на всеки фенотип чрез IF функции. ?matchType категоризира резултатите като “High Match” при 3 съвпадения, “Medium Match” при 2, “Low Match” при 1 съвпадение и “No Match” иначе. Чрез FILTER не се извеждат резултатите, които са “No Match”. Резултатите са сортирани в намаляващ ред по брой намерени ключови думи.

```

1 PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
2
3 SELECT ?phenotype ?phenotypeLabel ?matchType
4 WHERE {
5   ?phenotype rdfs:label ?phenotypeLabel .
6
7   BIND(
8     (IF(CONTAINS(LCASE(?phenotypeLabel), "cardiac"), 1, 0) +
9      IF(CONTAINS(LCASE(?phenotypeLabel), "muscle"), 1, 0) +
10     IF(CONTAINS(LCASE(?phenotypeLabel), "contraction"), 1, 0))
11    AS ?matchCount
12  )
13
14  BIND(
15    IF(?matchCount = 3, "High Match",
16     IF(?matchCount = 2, "Medium Match",
17      IF(?matchCount = 1, "Low Match", "No Match")))
18    AS ?matchType
19  )
20
21  FILTER(?matchCount > 0)
22 }
23 ORDER BY DESC(?matchCount)

```

Run

keyboard shortcuts

Filter query results Compact view ☐ Hide row numbers ☐ Showing results from 0 to 857 of 857. Query took 0.3s, moments ago.

|     | phenotype          | phenotypeLabel                                      | matchType      |
|-----|--------------------|-----------------------------------------------------|----------------|
| 1   | obo:GO_0055117     | "regulation of cardiac muscle contraction"          | "High Match"   |
| 2   | obo:GO_0060048     | "cardiac muscle contraction"                        | "High Match"   |
| 3   | obo:GO_0055118     | "negative regulation of cardiac muscle contraction" | "High Match"   |
| 4   | obo:GO_0060452     | "positive regulation of cardiac muscle contraction" | "High Match"   |
| 5   | obo:CL_0000513     | "cardiac muscle myoblast"                           | "Medium Match" |
| 6   | obo:UBERON_0001133 | "cardiac muscle tissue"                             | "Medium Match" |
| 7   | obo:CL_0000746     | "cardiac muscle cell"                               | "Medium Match" |
| 8   | obo:CL_2000046     | "ventricular cardiac muscle cell"                   | "Medium Match" |
| 100 | obo:CL_0002600     | "smooth muscle cell of trachea"                     | "Low Match"    |
| 101 | obo:CL_0002601     | "uterine smooth muscle cell"                        | "Low Match"    |
| 102 | obo:UBERON_0002081 | "cardiac atrium"                                    | "Low Match"    |
| 103 | obo:CL_0002673     | "tongue muscle cell"                                | "Low Match"    |
| 104 | obo:UBERON_0000378 | "tongue muscle"                                     | "Low Match"    |

### 3.8. Заявка, обединяваща неврологични и мускулни аномалии

Заявката обединява две части от йерархията на класовете, като извежда подкласовете на клас Abnormality of the nervous system (HP\_0000707) и тези на клас Abnormality of the musculature (HP\_0003011). Чрез BIND подкласовете на първия клас се отбелязват с категория “Neurological”, а на втория – с категория “Muscular”. Резултатите са сортирани по азбучен ред на категорията и на етикета на аномалията (фенотип).

Резултат без включен логически извод (Include inferred data in results: OFF):

```
1 * PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
2 * PREFIX obo: <http://purl.obolibrary.org/obo/>
3
4 * SELECT DISTINCT ?phenotype ?phenotypeLabel ?category
5 * WHERE {
6 *   {
7 *     ?phenotype rdfs:subClassOf obo:HP_0000707 ;
8 *     rdfs:label ?phenotypeLabel .
9 *     BIND("Neurological" AS ?category)
10 *   }
11 *   UNION
12 *   {
13 *     ?phenotype rdfs:subClassOf obo:HP_0003011 ;
14 *     rdfs:label ?phenotypeLabel .
15 *     BIND("Muscular" AS ?category)
16 *   }
17 * }
18 * ORDER BY ?category ?phenotypeLabel
```

Run  
keyboard shortcuts

Filter query results Compact view ☐ Hide row numbers ☐ Showing results from 0 to 7 of 7. Query took 0.1s, moments ago.

|   | phenotype      | phenotypeLabel                                 | category       |
|---|----------------|------------------------------------------------|----------------|
| 1 | obo:HP_0011804 | "Abnormal muscle physiology"                   | "Muscular"     |
| 2 | obo:HP_0011805 | "Abnormal skeletal muscle morphology"          | "Muscular"     |
| 3 | obo:HP_0040172 | "Abnormality of occipitofrontalis muscle"      | "Muscular"     |
| 4 | obo:HP_0009728 | "Neoplasm of striated muscle"                  | "Muscular"     |
| 5 | obo:HP_0012639 | "Abnormal nervous system morphology"           | "Neurological" |
| 6 | obo:HP_0012638 | "Abnormal nervous system physiology"           | "Neurological" |
| 7 | obo:HP_0410008 | "Abnormality of the peripheral nervous system" | "Neurological" |

Резултат с включен логически извод (Include inferred data in results: ON):

```
1 * PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
2 * PREFIX obo: <http://purl.obolibrary.org/obo/>
3
4 * SELECT DISTINCT ?phenotype ?phenotypeLabel ?category
5 * WHERE {
6 *   {
7 *     ?phenotype rdfs:subClassOf obo:HP_0000707 ;
8 *     rdfs:label ?phenotypeLabel .
9 *     BIND("Neurological" AS ?category)
10 *   }
11 *   UNION
12 *   {
13 *     ?phenotype rdfs:subClassOf obo:HP_0003011 ;
14 *     rdfs:label ?phenotypeLabel .
15 *     BIND("Muscular" AS ?category)
16 *   }
17 * }
18 * ORDER BY ?category ?phenotypeLabel
```

Run  
keyboard shortcuts

| Filter query results |                | Compact view <input type="checkbox"/> Hide row numbers <input type="checkbox"/> | Showing results from 0 to 1000 of 3429. Query took 0.3s, moments ago. |
|----------------------|----------------|---------------------------------------------------------------------------------|-----------------------------------------------------------------------|
|                      | phenotype      | phenotypeLabel                                                                  | category                                                              |
| 1                    | obo:HP_0009023 | "Abdominal wall muscle weakness"                                                | "Muscular"                                                            |
| 2                    | obo:HP_0020202 | "Abnormal Z disk morphology"                                                    | "Muscular"                                                            |
| 3                    | obo:HP_0040286 | "Abnormal axial muscle morphology"                                              | "Muscular"                                                            |
| 4                    | obo:HP_0001430 | "Abnormal calf musculature morphology"                                          | "Muscular"                                                            |
| 662                  | obo:HP_0032457 | "2-3-layered lissencephaly"                                                     | "Neurological"                                                        |
| 663                  | obo:HP_0006818 | "4-layered lissencephaly"                                                       | "Neurological"                                                        |
| 664                  | obo:HP_0032475 | "6-layered lissencephaly"                                                       | "Neurological"                                                        |
| 665                  | obo:HP_0012651 | "Abasia"                                                                        | "Neurological"                                                        |

Обяснения на причините за установените различия: Когато логическият извод е изключен, заявката връща само класовете, които са изрично свързани в онтологията със свойството `rdfs:subClassOf` към класа `Abnormality of the nervous system (HP_0000707)` или класа `Abnormality of the musculature (HP:0003011)`. Това води до 7 резултата. Когато логическият извод е включен, свойството `rdfs:subClassOf` е транзитивно, следователно заявката връща както директно свързаните с един от двата класове, така и тези класове, които са подкласове на някой подклас на двата класа. Това води до 3429 резултата.

### 3.9. Заявка, извеждаща аномалиите, които са едновременно мускулни и неврологични

Заявката извежда всички аномалии, които са едновременно подкласове на `Abnormality of the musculature (HP_0003011)` и на `Abnormality of the nervous system (HP_0000707)`. Резултатите са сортирани по азбучен ред на етикета на аномалията (фенотип).

Резултатът от заявката представлява сечение на двете множества `Abnormality of the musculature (HP_0003011)` и `Abnormality of the nervous system (HP_0000707)`. Тъй като в SPARQL няма вградена клауза за сечение (като например `INTERSECT` в MySQL), това се постига чрез `FILTER EXISTS`.



```

1 PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
2 PREFIX obo: <http://purl.obolibrary.org/obo/>
3
4 SELECT DISTINCT ?phenotype ?phenotypeLabel
5 WHERE {
6   ?phenotype rdfs:subClassOf obo:HP_0003011 ;
7             rdfs:label ?phenotypeLabel .
8
9   FILTER EXISTS {
10    ?phenotype rdfs:subClassOf obo:HP_0000707 .
11  }
12 }
13 ORDER BY ?phenotypeLabel

```

Run

keyboard shortcuts

Table Raw response Pivot Table Google Chart Download as

Filter query results Compact view ☐ Hide row numbers ☐ Showing results from 0 to 54 of 54. Query took 0.1s, moments ago.

|   | phenotype      | phenotypeLabel            |
|---|----------------|---------------------------|
| 1 | obo:HP_0034353 | "Appendicular spasticity" |
| 2 | obo:HP_0430025 | "Bilateral facial palsy"  |
| 3 | obo:HP_0001283 | "Bulbar palsy"            |
| 4 | obo:HP_0031866 | "Clasp-knife sign"        |

### 3.10. Заявка, извеждаща аномалиите, които са мускулни

Заявката извежда всички аномалии, които са подкласове на Abnormality of the musculature (HP\_0003011). Резултатите са сортирани по азбучен ред на етикета на аномалията (фенотип).

```

1 PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
2 PREFIX obo: <http://purl.obolibrary.org/obo/>
3 SELECT ?phenotype ?phenotypeLabel
4 WHERE {
5   ?phenotype rdfs:subClassOf obo:HP_0003011 ;
6             rdfs:label ?phenotypeLabel .
7 }
8 ORDER BY ?phenotypeLabel

```

Press Alt+Enter to autocomplete

Run

keyboard shortcuts

Table Raw response Pivot Table Google Chart Download as

Filter query results Compact view ☐ Hide row numbers ☐ Showing results from 0 to 661 of 661. Query took 0.1s, moments ago.

|   | phenotype      | phenotypeLabel                         |
|---|----------------|----------------------------------------|
| 1 | obo:HP_0009023 | "Abdominal wall muscle weakness"       |
| 2 | obo:HP_0020202 | "Abnormal Z disk morphology"           |
| 3 | obo:HP_0040286 | "Abnormal axial muscle morphology"     |
| 4 | obo:HP_0001430 | "Abnormal calf musculature morphology" |

### 3.11. Заявка, извеждаща аномалиите, които са мускулни, но не са неврологични

Заявката извежда всички аномалии, които са подкласове на Abnormality of the musculature (HP\_0003011), но не са подкласове на Abnormality of the nervous system (HP\_0000707). Резултатите са сортирани по азбучен ред на етикета на аномалията (фенотип).

Резултатът от заявката представлява разлика на двете множества Abnormality of the musculature (HP\_0003011) и Abnormality of the nervous system (HP\_0000707). Тъй като в SPARQL няма вградена клауза за разлика, това се постига чрез FILTER NOT EXISTS.

The screenshot shows a SPARQL query editor with the following query:

```
1 PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
2 PREFIX obo: <http://purl.obolibrary.org/obo/>
3 SELECT ?phenotype ?phenotypeLabel
4 WHERE {
5   ?phenotype rdfs:subClassOf obo:HP_0003011 ;
6             rdfs:label ?phenotypeLabel .
7
8   FILTER NOT EXISTS {
9     ?phenotype rdfs:subClassOf obo:HP_0000707 .
10  }
11
12 }
13 ORDER BY ?phenotypeLabel
```

Below the query editor, there are tabs for 'Table', 'Raw response', 'Pivot Table', and 'Google Chart'. The 'Table' tab is selected, showing a table with 2 columns: 'phenotype' and 'phenotypeLabel'. The table contains 4 rows of results:

|   | phenotype      | phenotypeLabel                         |
|---|----------------|----------------------------------------|
| 1 | obo:HP_0009023 | "Abdominal wall muscle weakness"       |
| 2 | obo:HP_0020202 | "Abnormal Z disk morphology"           |
| 3 | obo:HP_0040286 | "Abnormal axial muscle morphology"     |
| 4 | obo:HP_0001430 | "Abnormal calf musculature morphology" |

Анализ на резултатите: Резултатите от заявки 3.9., 3.10. и 3.11. могат да бъдат съпоставени, тъй като се извършват над еднакви множества. От 3.9. се вижда, че аномалиите, които са едновременно мускулни и неврологични ( $M \cup N$ ) са 54 на брой. От 3.10. се вижда, че аномалиите, които са само мускулни ( $M$ ) са 661 на брой. От 3.11. се вижда, че аномалиите които са мускулни, но не са неврологични ( $M \setminus N$ ) са 607 на брой.

Очаква се  $|M| - |M \setminus N| = |M \cup N|$ . Проверка:  $661 - 607 = 54$ .

### 3.12. Заявка, извеждаща всички фенотипове, които имат дефиниции, съдържащи “noise”

Заявката извежда подкласовете на класа Phenotypic abnormality (HP:0000118), които имат дефиниции, съдържащи “noise”, както и самите дефиниции. Дефинициите се намират чрез свойството definition (obo:IAO\_0000115). Резултатите са сортирани по азбучен ред на фенотиповете. Това позволява търсене по ключови думи не само в името на даден фенотип, а и в неговото описание.

```
1 PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
2 PREFIX obo: <http://purl.obolibrary.org/obo/>
3
4 SELECT ?phenotype ?phenotypeLabel ?phenotypeDefinition
5 WHERE {
6   ?phenotype rdfs:subClassOf obo:HP_0000118 ;
7             rdfs:label ?phenotypeLabel ;
8             obo:IAO_0000115 ?phenotypeDefinition .
9   FILTER(CONTAINS(LCASE(?phenotypeDefinition), "noise"))
10 }
11 ORDER BY ?phenotypeLabel
```

Run

keyboard shortcuts

Filter query results

Compact view ☐ Hide row numbers ☐

Showing results from 0 to 9 of 9. Query took 0.2s, moments ago.

|   | phenotype      | phenotypeLabel             | phenotypeDefinition                                                                                                                                                                                                                                         |
|---|----------------|----------------------------|-------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------|
| 1 | obo:HP_0031657 | "Abnormal heart sound"     | "Any abnormal noise generated by the beating heart."                                                                                                                                                                                                        |
| 2 | obo:HP_5200062 | "Auditory sensory seeking" | "Pursuit of specific sounds or noises that is abnormal in intensity and/or frequency."                                                                                                                                                                      |
| 3 | obo:HP_0034424 | "Clicking tinnitus"        | "A type of tinnitus that presents as clicks, resembling the noise made by the snapping together of 2 fingers."                                                                                                                                              |
| 4 | obo:HP_5200290 | "Exploding head syndrome"  | "Sleep disorder in which one hears a loud noise or explosive crashing sound in their head. The sound is not real or heard by anyone else. The episode typically happens suddenly, either when beginning to fall asleep or when waking up during the night." |

### 3.13. Заявка, извеждаща всички фенотипове, които имат коментари

Заявката извежда подкласовете на класа Phenotypic abnormality (HP:0000118), които имат коментари. Коментарите се намират чрез свойството `rdfs:comment`. Резултатите са сортирани по азбучен ред на фенотиповете.

```
1 * PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
2 PREFIX obo: <http://purl.obolibrary.org/obo/>
3
4 SELECT ?phenotype ?phenotypeLabel ?phenotypeComment
5 * WHERE {
6   ?phenotype rdfs:subClassOf obo:HP_0000118 ;
7             rdfs:label ?phenotypeLabel ;
8             rdfs:comment ?phenotypeComment .
9 }
10 ORDER BY ?phenotypeLabel
```

  
  
keyboard shortcuts

Filter query results Compact view ☐ Hide row numbers ☐ Showing results from 0 to 1000 of 4134. Query took 0.1s, moments ago.

|   | phenotype      | phenotypeLabel                 | comment                                                                                                                                                                                         |
|---|----------------|--------------------------------|-------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------------|
| 1 | obo:HP_0000878 | "11 pairs of ribs"             | "Normally, there are 12 pairs of ribs."                                                                                                                                                         |
| 2 | obo:HP_0025770 | "2-3 finger osseus syndactyly" | "Fusion of the second and third finger, involving soft parts and including fusion of individual finger bones."                                                                                  |
| 3 | obo:HP_0020147 | "2-Methylbutyryl glycinuria"   | "This phenotype can be related to a defect in the degradation pathway of L- isoleucine leading to increased urinary excretion of 2-methylbutyryl glycine."                                      |
| 4 | obo:HP_0033220 | "2-ethylhydracrylic aciduria"  | "Isolated excretion of 2-ethylhydracrylic acid is the hallmark of short/branched-chain acyl-CoA dehydrogenase deficiency (SBCADD), a defect in the proximal pathway of L-isoleucine oxidation." |

### 3.14. Заявка, извеждаща фенотиповете, които имат свойство has part

Заявката, извежда всички фенотипове, които са свързани с други обекти чрез свойството `has part` (obo:BFO\_0000051) в рамките на OWL рестрикции. Всеки фенотип, който има етикет (`rdfs:label`) и участва в рестрикция с `owl:onProperty` към `obo:BFO_0000051` и `owl:someValuesFrom` към определена стойност, се извлича заедно със своя етикет. Резултатите са уникални (чрез `DISTINCT`) и са сортирани по азбучен ред на фенотиповете.

Резултат без включен логически извод (Include inferred data in results: OFF):

```

1 PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
2 PREFIX owl: <http://www.w3.org/2002/07/owl#>
3 PREFIX obo: <http://purl.obolibrary.org/obo/>
4
5 SELECT DISTINCT ?phenotype ?phenotypeLabel
6 WHERE {
7   ?phenotype rdfs:label ?phenotypeLabel ;
8             rdfs:subClassOf ?owlRestriction .
9   ?owlRestriction owl:onProperty obo:BF0_0000051 ;
10                  owl:someValuesFrom ?restrictionValue .
11 }
12 ORDER BY ?phenotypeLabel

```

Run keyboard shortcuts

Table Raw response Pivot Table Google Chart Download as

Filter query results Compact view ☐ Hide row numbers ☐ Showing results from 0 to 457 of 457. Query took 0.1s, moments ago.

|   | phenotype          | phenotypeLabel                                                                                                     |
|---|--------------------|--------------------------------------------------------------------------------------------------------------------|
| 1 | obo:CHEBL47909     | "3-oxo-Delta(4) steroid"                                                                                           |
| 2 | obo:GO_0007186     | "G protein-coupled receptor signaling pathway"                                                                     |
| 3 | obo:CHEBL18259     | "N-acetyl-beta-D-galactosaminyl-(1->3)-alpha-D-galactosyl-(1->4)-beta-D-galactosyl-(1->4)-beta-D-glucosylceramide" |
| 4 | obo:UBERON_0010543 | "acropodial skeleton"                                                                                              |

Резултат с включен логически извод (Include inferred data in results: ON):

```

1 PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
2 PREFIX owl: <http://www.w3.org/2002/07/owl#>
3 PREFIX obo: <http://purl.obolibrary.org/obo/>
4
5 SELECT DISTINCT ?phenotype ?phenotypeLabel
6 WHERE {
7   ?phenotype rdfs:label ?phenotypeLabel ;
8             rdfs:subClassOf ?owlRestriction .
9   ?owlRestriction owl:onProperty obo:BF0_0000051 ;
10                  owl:someValuesFrom ?restrictionValue .
11 }
12 ORDER BY ?phenotypeLabel

```

Run keyboard shortcuts Press Alt+Enter to autocomplete

Table Raw response Pivot Table Google Chart Download as 1 2 3 4 5

Filter query results Compact view ☐ Hide row numbers ☐ Showing results from 0 to 1000 of 25290. Query took 1.6s, minutes ago.

|   | phenotype      | phenotypeLabel                     |
|---|----------------|------------------------------------|
| 1 | obo:CL_0019031 | "intestine goblet cell"@en         |
| 2 | obo:CL_0008002 | "skeletal muscle fiber"@en         |
| 3 | obo:CL_0008008 | "striated visceral muscle cell"@en |
| 4 | obo:CL_0019003 | "tracheobronchial goblet cell"@en  |

Обяснения на причините за установените различия: Когато логическият извод е изключен, заявката връща само фенотиповете, които имат изрично дефинирани OWL рестрикции в онтологията със свойството `has part` (`obo:BFO_0000051`). Това ограничава резултатите до случаите, където връзката е директно аотирана. Това води до 457 резултата. Когато логическият извод е включен, се позволява извличането на допълнителни фенотипове, които са свързани индиректно чрез транзитивност или други логически конструкции в онтологията - например чрез йерархия на класовете или на свойствата. Това води до 25290 резултата.

#### 4. Заключение и бъдещо развитие

Проектът демонстрира възможностите на Human Phenotype Ontology чрез разработените заявки за извличане и анализ на фенотипна информация. Изследвани са част от свойствата на фенотиповете, показани са разнообразни начини за търсене по ключови думи и за извличане на връзки между различни фенотипове.

Проектът може да се разшири чрез разглеждане на други подйерархии на класовете като Mode of inheritance (HP:0000005) и Clinical course (HP:0031797), като и чрез интегриране на HPO с други онтологии, съдържащи информация за гени и заболявания.

#### 5. Използвани технологии

- **SPARQL** – за реализация на заявките
- **GraphDB** – за изпълняване на заявките и визуализиране на резултатите им; за визуализиране на онтологията

#### 6. Файлова структура на проекта

- **documentation\_8MI3400591.pdf** – документация на проекта
- **hp.owl** – Human Phenotype Ontology, v2025-05-06
- **SPARQL\_queries\_8MI3400591.txt** – код на заявките

#### 7. Източници

- [1] Информация за Human Phenotype Ontology: <https://hpo.jax.org/>
- [2] Изтегляне на Human Phenotype Ontology, v2025-05-06: <https://hpo.jax.org/data/ontology>
- [3] Документация на GraphDB: <https://graphdb.ontotext.com/documentation/11.0/>
- [4] Документация на SPARQL: <https://www.w3.org/TR/sparql11-query/>